

AMIGOS DEL **MSX**

AÑO II

N.º 7

**295
PTS.**

- CLASIFICA
- CUADRADO MALDITO
- MONZA
- MARATHON
- ATAQUE
- ESQUI
- DEFENDER
- RALLY
- SIETE Y MEDIA



TODO SOBRE EL

N.º 15 690 Ptas.

AMSTRAD

LASER




AMIGOS DEL MSX




SUMARIO

- CLASIFICA
- CUADRADO MALDITO
- MONZA
- MARATHON
- ATAQUE
- ESQUI
- DEFENDER
- RALLY
- SIETE Y MEDIA



Edita: Editorial GTS. C/. Bailén, 20. 1.º Izda. 28005 MADRID. **Secretaría Redacción:** Margarita Rancero. **Colaboradores:** Eugenio Garrido, J.F. Martínez, J. Bernal, R. Carralón, J. Ramos, Juan Jesús Ortega. **Dirección Artística y Técnica:** Carlos Gorrindo. **Publicidad:** Bailén, 20. 1.º. 28005 MADRID. **Imprime:** GREFOL, S. A., Pol. II, La Fuensanta, Móstoles (Madrid). **Distribuye:** R.B.A. Promotora de Ediciones, S. A. Trav. de Gracia, 56. Atico 1.ª. Teléfono: 200 82 56. **Depósito Legal:** M. 3.988-1986.



EXPLICACION DE LOS PROGRAMAS

CLASIFICA

```
10 ' PROGRAMA
20 ' CLASIFICACION POR EL
30 '
40 KEYOFF:WIDTH39
50 COLOR15,4,4:SCREEN0
60 LOCATE5,1:PRINT"SISTEMAS DE CLASIFICACION"
70 LOCATE0,5:PRINT"Los ordenadores a la hora de clasificar un conjunto de palabra
s, evalúan cada carácter. Sin embargo, como no conocen el alfabeto, en esta eva
luación se guían por los códigos ASCII que no son exactamente iguales que el
alfabeto."
80 PRINT"Para el ordenador no será lo mismo <a> que <a> ya que la primera tiene
dis- tinto código que la segunda; de igual"
90 PRINT"forma, los caracteres en mayúsculas no tienen el mismo código y la ñ no
tiene el código entre la n y la o, por lo que a la hora de hacer una clasificaci
ón, deberemos tener en cuenta todos estos factores."
100 LOCATE2,22:PRINT"Pulsa una tecla para continuar"
110 IF INKEY$="" THEN 110 ELSE CLS
120 LOCATE7,0:PRINT"METODO DE MINIMOS"
130 PRINT:PRINT" Este mtodo de clasificación es el":PRINT:PRINT"más lento en ej
ecución pero el más ":PRINT:PRINT"simple en estructura. Consiste en la":PRINT:PR
INT"comparación de una palabra con todas":PRINT:PRINT"las restantes de la lista,
de manera"
```



```

140 PRINT:PRINT"que sustituye la palabra comparadora,":PRINT:PRINT"por aquella q
ue presenta el mínimo":PRINT
150 PRINT"valor de nuestra lista; a continuación":PRINT:PRINT"compara la siguien
te y así sucesiva-":PRINT:PRINT"mente hasta completar la lista."
160 LOCATE2,22:PRINT"Pulsa una tecla para continuar"
170 IF INKEY$="" THEN 170 ELSE CLS
180 LOCATE5,0:PRINT"SUBROUTINA DE CLASIFICACION"
190 LOCATE9,2:PRINT"METODO DE MINIMOS"
200 LOCATE0,4:PRINT" 10 FOR i=1 TO n-1":PRINT:PRINT" 20 m$=a$(i):k=i":PRINT:PRIN
T" 30 FOR j=i+1 TO n":PRINT:PRINT" 40 IF a$(j)<m$ THEN k=j:m$=a$(j)":PRINT
210 PRINT" 50 NEXT j":PRINT:PRINT" 60 SWAP a$(i),a$(k)":PRINT:PRINT" 70 NEXT i":
PRINT:PRINT" n número de datos a clasificar":PRINT:PRINT" a$(n) matriz que almac
ena los datos"
215 LOCATE2,22:PRINT"Pulsa una tecla para continuar"
220 IF INKEY$="" THEN 220
230 SCREEN0:CLEAR 2000
240 PRINT"introduce 10 palabras de 5 letras como máximo"
250 DIM P(20):N=10:PL=11:K=18
260 FOR Z=1 TO N
270 PRINT"palabra número";Z;
280 INPUT P$(Z)
290 IF LEN(P$(Z))>5 THEN 270
300 NEXT
310 SCREEN 2,2
320 OPEN"grp:" AS #1
330 H=180
340 DIM A(1,15),T$(20)
350 FOR Z=1 TO 8
360 D$=D$+CHR$(0)
370 NEXT
380 PL=PL-1
390 FOR Z=56 TO 87 STEP 6
400 PC=(Z-56)/6+1
410 A$=MID$(P$(PL),PC,1)
420 DRAW"bm=z:,0"
430 PRINT#1,A$
440 NEXT
450 FOR D=56 TO 87
460 IF D<72 THEN A(0,D-56)=VPEEK(D) ELSE A(1,D-72)=VPEEK(D)
470 NEXT D
480 FOR V=1 TO 2
490 FOR X=0 TO 15
500 T$(K)=T$(K)+CHR$(A(0,X))
510 T$(K+1)=T$(K+1)+CHR$(A(1,X))
520 IF X=7 OR X=15 THEN T$(K)=T$(K)+D$:T$(K+1)=T$(K+1)+D$
530 NEXT
540 NEXT
550 SPRITE$(K)=T$(K)
560 SPRITE$(K+1)=T$(K+1)
570 PUT SPRITE K, (56,-1),15,K
580 PUT SPRITE K+1, (72,-1),15,K+1
590 LINE(0,0)-(255,10),4,BF
600 FOR Y=-1 TO H
610 PUT SPRITE K, (56,Y),15,K
620 PUT SPRITE K+1, (72,Y),15,K+1
630 SOUND0,Y+70:SOUND1,K/2+1
640 SOUND8,6
650 NEXT

```



```

660 SOUND8,0
670 P(K)=H
680 K=K-2
690 H=H-18
700 IF K>=0 THEN 380
710 DRAW"bm145,0":PRINT#1,"CLASIFICACION":DRAW"BM145,10":PRINT#1,"POR EL METODO"
:DRAW"BM157,20":PRINT#1,"DE MINIMOS"
720 FOR I=1 TO N-1
730 M#=P$(I)
740 DRAW"bm150,40":PRINT#1,"Comparando"
750 DRAW"bm170,60":PRINT#1,P$(I)
760 GOSUB 960
770 L=I
780 FOR J=I+1 TO N
790 LINE(150,100)-(255,110),4,BF:DRAW"bm174,80":PRINT#1,"con ":DRAW"BM170,100":P
RINT#1,P$(J)
800 GOSUB 1040
810 IF P$(J)<M# THEN L=J:M#=P$(J)
820 LINE(205,130)-(255,140),4,BF:DRAW"bm150,130":PRINT#1,"Mínimo: ";M#
830 DRAW"BM150,150":PRINT#1,"Para seguir":DRAW"bm150,170":PRINT#1,"pulsa tecla"
840 IF INKEY$="" THEN 840 ELSE BEEP
850 NEXT
860 GOSUB 1120
870 SWAP P$(I),P$(L)
880 IF P$(I)<>P$(L) THEN GOSUB 1180 ELSE GOSUB 1640
890 NEXT
900 GOSUB 1150
910 DRAW"bm120,50":PRINT#1,"Fin de la":DRAW"BM120,70":PRINT#1,"clasificación."
920 DRAW"bm120,90":PRINT#1,"Pulsa P para otra":DRAW"bm120,110":PRINT#1,"clasific
ación"
930 DRAW"bm120,130":PRINT#1,"Pulsa T para":DRAW"bm120,150":PRINT#1,"terminar"
940 R$=INPUT$(1)
950 IFR$="P" OR R$="p" THEN RUN 230 ELSE IF R$="t" OR R$="T" THEN SCREEN0:LOCATE
4,10:PRINT"PARA EL SIGUIENTE PROGRAMA":LOCATE4,12:PRINT"PULSA PLAY EN EL CASSET
TE":CLOAD ELSE 940
960 REM cambio color sprite i
970 II=(I-1)*2
980 IF II<1 THEN 1010
990 PUT SPRITE II-2,(56,P(II-2)),15,II-2
1000 PUT SPRITE II-1,(72,P(II-2)),15,II-1
1010 PUT SPRITE II,(56,P(II)),6,II
1020 PUT SPRITE II+1,(72,P(II)),6,II+1
1030 RETURN
1040 REM cambio color sprite j
1050 JJ=(J-1)*2
1060 IF JJ-2=II THEN 1090
1070 PUT SPRITE JJ-2,(56,P(JJ-2)),15,JJ-2
1080 PUT SPRITE JJ-1,(72,P(JJ-2)),15,JJ-1
1090 PUT SPRITE JJ,(56,P(JJ)),6,JJ
1100 PUT SPRITE JJ+1,(72,P(JJ)),6,JJ+1
1110 RETURN
1120 PUT SPRITE JJ,(56,P(JJ)),15,JJ
1130 PUT SPRITE JJ+1,(72,P(JJ)),15,JJ+1
1140 RETURN
1150 PUT SPRITE II,(56,P(II)),15,II
1160 PUT SPRITE II+1,(72,P(II)),15,II+1
1170 RETURN
1180 REM cambio posición sprites
1190 LINE(150,40)-(255,191),4,BF

```




```

1200 DRAW"bm160,50":PRINT#1,"Cambio: ":DRAW"bm160,70":PRINT#1,P$(L)
1210 DRAW"bm170,90":PRINT#1,"por":DRAW"bm160,110":PRINT#1,P$(I)
1220 FOR T=1 TO 500:NEXT
1230 LL=(L-1)*2:CC=56
1240 SOUND1,1:SOUND8,6
1250 FOR X=56 TO 102
1260 CC=CC-1
1270 PUTSPRITELL,(X,P(LL)),3,LL
1280 PUTSPRITELL+1,(X+16,P(LL)),3,LL+1
1290 PUTSPRITEII,(CC,P(II)),3,II
1300 PUTSPRITEII+1,(CC+16,P(II)),3,II+1
1310 SOUND0,X
1320 NEXT X
1330 YY=P(LL)
1340 FOR Y=P(II) TO P(LL)
1350 YY=YY-1
1360 PUTSPRITELL,(102,YY),3,LL
1370 PUTSPRITELL+1,(118,YY),3,LL+1
1380 PUTSPRITEII,(10,Y),3,II
1390 PUTSPRITEII+1,(26,Y),3,II+1
1400 SOUND0,Y
1410 NEXT
1420 CC=10
1430 FOR X=102 TO 56 STEP-1
1440 CC=CC+1
1450 PUTSPRITELL,(X,P(II)),3,LL
1460 PUTSPRITELL+1,(X+16,P(II)),3,LL+1
1470 PUTSPRITEII,(CC,P(LL)),3,II
1480 PUTSPRITEII+1,(CC+16,P(LL)),3,II+1
1490 SOUND0,X
1500 NEXT X
1510 SOUND8,0
1520 SPRITE$(II)=T$(LL)
1530 SPRITE$(LL)=T$(II)
1540 SPRITE$(II+1)=T$(LL+1)
1550 SPRITE$(LL+1)=T$(II+1)
1560 PUTSPRITEII,(56,P(II)),15,II
1570 PUTSPRITEII+1,(72,P(II)),15,II+1
1580 PUTSPRITEII,(56,P(LL)),15,LL
1590 PUTSPRITEII+1,(72,P(LL)),15,LL+1
1600 SWAP T$(II),T$(LL)
1610 SWAP T$(II+1),T$(LL+1)
1620 LINE(150,40)-(255,191),4,BF
1630 RETURN
1640 REM no hay cambio de sprites
1650 LINE(150,40)-(255,191),4,BF
1660 DRAW"bm150,50":PRINT#1,"No hay cambio":DRAW"bm150,70":PRINT#1,"por estar ya
":DRAW"bm150,90":PRINT#1,"clasificado"
1670 FOR T=1 TO 1000:NEXT
1680 LINE(150,40)-(255,191),4,BF
1690 RETURN

```





CUADRADO MALDITO


```
10 REM PROGRAMA <CUADRADO MALDITO>
20 "
30 COLOR 15,4,4:SCREEN 0:WIDTH 30:KEYOFF
40 PRINTTAB(7);"CUADRADO MALDITO":PRINT:PRINT" El cuadrado está definido por
un conjunto de números que tu debes averiguar, todas las filas, columnas y diagonales
suman igual."
50 PRINT:PRINT" Los cuadrados que tu puedes definir tienen número impar de casil-
las y todos los números que contienen son enteros y están formados, como máxi-
mo, por dos cifras."
60 PRINT:PRINT" La única condición es que los números no se repitan. ":PRI
NT" Introduce las letras ZZ si quieres abandonar o presentar tu jugada."
70 PRINT:PRINT"PULSA UNA TECLA PARA COMENZAR"
80 IFINKEY$=""THENB0
90 SCREEN0:WIDTH39
100 DIMC(9,9),B(9,9)
105 PRINT:PRINTTAB(9);"ESCRIBE EN MAYUSCULAS"
110 LOCATE0,10:INPUT"ELIGE EL TAMAÑO DEL CUADRADO (ENTRE 1 Y 9 E IMPAR
) ";B
120 IFB/2=INT(B/2)ORB>9ORB<1ORB<>INT(B)THENPLAY"02CC":CLS:GOTO105
130 CLS:LOCATE 0,10:PRINT"UN MOMENTO, POR FAVOR, ESTOY PENSANDO.."
140 E=INT(RND(-TIME)*20)
150 IFE<0ORE>99THEN140
160 D=E:A=1:I=1:J=(B+1)/2
170 C(I,J)=D:D=D+1:IFD>B*B+E-1THEN260
180 IFA<BTHEN200
190 A=1:I=I+1:GOTO170
200 A=A+1
210 I=I-1:J=J+1
220 IFI<>0THEN240
230 I=B:GOTO170
240 IFJ<=BTHEN170
250 J=1:GOTO170
260 D=((B*B*B+B)/2)+B*(E-1)
270 SCREEN2:OPEN"GRP:"AS#1
280 LINE(0,0)-(255,10),1,BF:DRAW"BM51,2":PRINT#1,"ADIVINA LOS NUMEROS"
290 FORY=25TOB*15+25STEP15
300 LINE(20,Y)-(B*24+16,Y+5),8,BF
310 NEXT
320 FORX=16TOB*24+16STEP24
330 LINE(X,25)-(X+7,B*15+30),9,BF
340 NEXT
350 CL=64:COLOR9:FORX=30TOB*24+18STEP24:CL=CL+1:DRAW"BM=X;,15":PRINT#1,CHR$(CL):
NEXT
360 CN=0:FORY=32TOB*15+22STEP15:CN=CN+1:DRAW"BM9,=Y;":PRINT#1,MID$(STR$(CN),2,2)
:NEXT
370 COLOR10:DRAW"BM20,=Y;":PRINT#1,"SUMAN: ";D
```



```


380 COLOR15:T$="Z" PARA TERMINAR":FORZ=1TOLEN(T$):Y=Z*9+16:DRAW"BM242,=Y;":PRINT
#1,MID$(T$,Z,1):NEXT
390 DRAW"BM20,182":PRINT#1,"¿LETRA Y NUMERO?":A$=INPUT$(2):IFA$=""THEN390
400 IFA$="ZZ"ORA$="zz"THEN600
410 F$=LEFT$(A$,1):N$=RIGHT$(A$,1)
420 IFASC(F$)>CLORASC(F$)<65THENPLAY"02C":GOTO390
430 IFASC(N$)>ASC(MID$(STR$(CN),2,1))ORASC(N$)<49THENPLAY"02C":GOTO390
440 Y=VAL(N$):X=ASC(F$)-64
450 YY=17+15*Y:XX=1+X*24
460 LINE(XX-1,YY-1)-(XX+14,YY+7),3,BF
470 LINE(0,180)-(255,191),4,BF
480 COLOR15:DRAW"BM10,182":PRINT#1,"ESCRIBE EL VALOR Y PULSA ENTER":B$=""
490 L$=INKEY$:IFL$=""THEN490
500 IFL$<>CHR$(13)THENB$=B$+L$:GOTO490ELSEIFB$=""THENPLAY"02C":GOTO480
510 Z=ASC(LEFT$(B$,1)):IFZ<48ORZ>57THENPLAY"02C":GOTO480
520 B(Y,X)=VAL(B$):IFB(Y,X)<0ORB(Y,X)>99ORB(Y,X)<>INT(B(Y,X))THENPLAY"02C":GOTO4
80
525 LINE(XX-1,YY-1)-(XX+14,YY+7),4,BF
530 DRAW"BM=XX;=YY;":PRINT#1,USING"##";B(Y,X):LINE(0,182)-(255,191),4,BF:GOTO39
0
600 LINE(0,182)-(255,191),4,BF
610 DRAW"BM20,182":PRINT#1,"ABANDONAS (A) O EVALUO (E)":R$=INPUT$(1)
620 IFR$<>"A"ANDR$<>"a"ANDR$<>"E"ANDR$<>"e"THENPLAY"02C":GOTO610
630 IFR$="A"ORR$="a"THEN1000
700 IFB=1THEN720
705 MX=0:MN=99:FORI=1TOB:FORJ=1TOB:IFMX<B(I,J)THENMX=B(I,J)ELSEIFMN>B(I,J)THENMN
=B(I,J)
710 NEXT:IFMX=MNTHEN1100
720 FORI=1TOB:FORJ=1TOB:X(I)=X(I)+B(I,J):Y(I)=Y(I)+B(J,I):NEXTJ,I
740 D1=0:D2=0:FORI=1TOB:D1=D1+B(I,I):NEXT:I=B:FORI=1TOB:D2=D2+B(I,I):J=J-1:NEXT
750 IFD1<>DORD2<>DTHEN1100
760 FORI=1TOB:IFX(I)<>DORY(I)<>DTHENI=B*3-1
770 NEXT:IFI=B*3THEN1100
800 ERASEX,Y:J=40:FORI=0TO191STEP8:PLAY"L64N=J;":LINE(0,I)-(255,I+7),8,BF:DRAW"B
M10,=I;":PRINT#1,"ENHORABUENA! LO HAS CONSEGUIDO":J=J+1:NEXT:FORT=1TO1000:NEXT
900 SCREEN2:DRAW"BM40,90":PRINT#1,"¿OTRA PARTIDA? (S/N)":R$=INPUT$(1):IFR$<>"S"A
NDR$<>"s"ANDR$<>"N"ANDR$<>"n"THEN900
910 IFR$="S"ORR$="s"THENRUN90ELSE1500
1000 FORI=1TOB:FORJ=1TOB
1010 YY=17+15*I:XX=1+J*24
1020 LINE(XX-1,YY-1)-(XX+14,YY+7),4,BF
1030 DRAW"BM=XX;=YY;":PRINT#1,USING"##";C(I,J)
1040 NEXTJ,I
1050 LINE(0,182)-(255,191),4,BF:DRAW"BM20,182":PRINT#1,"PULSA TECLA PARA CONTINU
AR"
1060 IFINKEY$=""THEN1060ELSE900
1100 ERASEX,Y:LINE(0,182)-(255,191),4,BF
1110 DRAW"BM20,182":PRINT#1,"EL CUADRADO NO ES CORRECTO":PLAY"02CCC":FORT=1TO100
0:NEXT
1120 LINE(0,182)-(255,191),4,BF
1130 DRAW"BM20,182":PRINT#1,"ABANDONAS (A) O CONTINUAS (C)":R$=INPUT$(1):IFR$<>"
A"ANDR$<>"a"ANDR$<>"C"ANDR$<>"c"THEN1130
1140 IFR$="A"ORR$="a"THEN1000
1150 LINE(0,182)-(255,191),4,BF
1160 GOTO390
1500 SCREEN0:WIDTH39
1510 LOCATE4,10:PRINT"PARA EL SIGUIENTE PROGRAMA":LOCATE4,12:PRINT"PULSA PLAY EN
EL CASSETTE"
1520 CLOAD
1530 END

```

MONZA

```
10 '000000000000000000000000
20 '0000      0000
30 '0000  MONZA  0000
40 '0000      0000
50 '000000000000000000000000
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '      1986
100 '000000000000000000000000
110 '
120 COLOR 15,13,1
130 SCREEN 3
140 OPEN"GRP:"AS1
150 PSET(40,80),12:PRINT#1,"MONZA"
160 FOR I=1 TO 1500
170 NEXT
180 CLOSE
190 COLOR 1,4,3
200 SCREEN 2, 2 :FOR K=1 TO 2 :END$="":FORI=1TO 32 :READ A$
205 B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRITE$ (K)=B$:NEXT
210 OPEN"grp:"AS1
220 LINE (0,0)-(255,17),12,BF
230 LINE (0,175)-(255,192),12,BF
240 PSET(75,70),4:PRINT#1,"!! LISTOS !!"
250 FOR I=1 TO 700:NEXT
260 LINE (75,70)-(225,78),4,BF
270 SOUND 6,15:SOUND 7,7:SOUND 8,16:SOUND 9,16:SOUND 10,16
275 SOUND 11,0 :SOUND 12,16:SOUND 13,0
280 K=127
290 TIME=0
300 G=289
310 ON STRIGGOSUB 580
320 STRIG(CT) ON
```


MARATHON

```

10 '00000000000000000000
20 '0000      0000
30 '0000 MARATON 0000
40 '0000      0000
50 '00000000000000000000
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '      1986
100 '00000000000000000000
110 '
120 COLOR 15,12,1:KEY OFF
130 SCREEN 3
140 OPEN"GRP:"AS1
150 PSET(25,80),12:PRINT#1,"MARATON"
160 FOR I=1 TO 1500
170 NEXT
180 CLOSE
190 COLOR 1,7,1
200 SCREEN 2, 2
210 OPEN"grp:"AS1
220 DEFINT X,Y,J,K
230 FOR K=1 TO 5 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$ (K)=B$:NEXT
240 FOR K=6 TO 7 :B$="":FORI=1TO 8 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRITE
$ (K)=B$:NEXT
250 FOR K=8 TO 9 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$ (K)=B$:NEXT
260 FOR K=10 TO 11:B$="":FORI=1TO 8 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$ (K)=B$:NEXT
270 G=260
280 AA=4:BA=8:C=5:D=9:E=6:F=10:GA=7:H=11
290 X1=1:Y1=165:X2=4:Y2=173:X3=19:Y3=172:X4=9:Y4=170
300 GOSUB 500
310 CIRCLE(50,35),15,10
320 PAINT(50,35),10
330 LINE (0,80)-(255,170),15,BF
340 LINE (0,170)-(255,192),4,BF
350 PSET(80,72),7:PRINT#1,"!!! LISTO !!!"
360 FOR I=1 TO 900:NEXT
370 LINE (75,70)-(225,78),7,BF
380 SOUND 6,15:SOUND 7,7:SOUND 8,16:SOUND 9,16:SOUND 10,16:SOUND 11,0 :SOUND 12,
16:SOUND 13,0

```


1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57 58 59 60 61 62 63 64 65 66 67 68 69 70 71 72 73 74 75 76 77 78 79 80 81 82 83 84 85 86 87 88 89 90 91 92 93 94 95 96 97 98 99 100



ATAQUE

```
10 '00000000000000000000
20 '0000 0000
30 '0000 ATAQUE 0000
40 '0000 0000
50 '00000000000000000000
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 ' 1986
100 '00000000000000000000
110 '
120 COLOR 1,4,6
130 KEY OFF:COLOR 15,1,1
140 SCREEN 2,3
150 VV=1
160 GOSUB 390
170 OPEN"GRP:"AS#1
180 PUT SPRITE 1,(83,76),10,1
190 PSET(140,90):PRINT#1,"ATAQUE"
200 LINE(15,178)-(240,192),3,BF
210 PSET(25,181):PRINT#1,"LUIS SANGUINO ARIAS - 1986"
220 FOR I= 1 TO 500:A=A+1:COLOR A-1,A+1,A
230 IF A>13 THEN A=1
240 NEXT
250 CLS:CLOSE
260 COLOR 15,1,1
270 SCREEN 1
280 LOCATE 3,10:PRINT"JOYSTICK O-CURSOR [J/C]"
290 X$=INKEY$
300 IF X$="j" OR X$="J" THEN CT=1:GOTO 360
310 IF X$="c" OR X$="C" THEN CT=0:GOTO 360
320 R=R+1
330 IF R>14 THEN R=1
340 VDP(7)=R
350 GOTO 290
360 OPEN"GRP:"AS 1
370 SCREEN 2,2
380 RESTORE
390 CO=1
400 H=RND(-TIME)
410 FOR W=1 TO 260
420 CO=CO+1:IF CO>15 THEN CO=2
```




```

430 K=INT(RND(1)*192)
440 PSET(W,K),CO
450 NEXT
460 FOR K=1 TO 2 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$ (K)=B$:NEXT
470 IF VV=1 THEN VV=2:RETURN
480 SPRITE ON
490 ON SPRITE GOSUB 750
500 X=125
510 GOSUB 710
520 Y1=Y1+9:Y2=Y2+8:Y3=Y3+9:Y8=Y8+10
530 Y4=Y4+9:Y5=Y5+8:Y6=Y6+9:Y7=Y7+10
540 PUTSPRITE 1,(X,175-RR),10,1
550 GOSUB 660
560 PUTSPRITE 2,(5+AL,Y1),2,2
570 PUTSPRITE 3,(40+AL,Y2),6,2
580 PUTSPRITE 4,(80+AL,Y3),7,2
590 PUTSPRITE 5,(115+AL,Y4),3,2
600 PUTSPRITE 6,(150+AL,Y5),13,2
610 PUTSPRITE 7,(185+AL,Y6),4,2
620 PUTSPRITE 8,(220+AL,Y7),2,2
630 PUTSPRITE 9,(250+AL,Y8),15,2
640 IF Y2>192 THEN GOSUB 710
650 GOTO 520
660 D=STICK(CT)
670 IF D=3 THEN X=X+6MOD256
680 IF D=7 THEN X=X-6MOD256
690 IF RR>160 THEN EX=1:GOTO 880
700 RETURN
710 AL=INT(RND(1)*25)
720 :Y2=0:RR=RR+5:Y8=-15
730 Y1=4:Y3=-16:Y4=-2:Y5=-9:Y6=0 :Y7=0
740 RETURN
750 SOUND 8,8
760 FOR Z=0 TO 255
770 SOUND 0,Z
780 NEXT
790 SOUND 8,0
800 SOUND 7,7
810 FOR Z=15 TO 0 STEP -1
820 COLOR,,15-Z
830 PUTSPRITE 1,(X,175-RR),Z,1
840 SOUND 8,Z
850 FOR T=1 TO 80
860 NEXT
870 NEXT
880 COLOR 1,14,12:SCREEN 1
890 IF EX=1 THEN 930
900 LOCATE 1,5 :PRINT"LO SIENTO HAS SIDO DESTRUIDO"
910 LOCATE 1,10 :PRINT"HAS RECORRIDO";RR*10000;"KM"
920 IF EX=0 THEN 940
930 LOCATE 0,7:PRINT"ENHORABUENA LO CONSEGUISTES"
940 LOCATE 1,20 :PRINT" OTRA PARTIDA SI O NO [S/N]"

```



ESQUI

```

10 '00000000000000000000
20 '0000      ,      0000
30 '0000      SKY      0000
40 '0000      0000
50 '00000000000000000000
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '      1986
100 '00000000000000000000
110 '
120 COLOR 10,2,1
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(45,80),2:PRINT#1," SKY"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOYSTICK 0 CURSOR{J/C}"
210 X$=INKEY$
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X$="J"OR X$="j" THEN CT=1:GOTO270
250 IF X$="C"OR X$="c" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 1,15,1:SCREEN 2, 2 :KEY OFF
280 FOR K=1 TO 5 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$ (K)=B$:NEXT
290 LINE (0,0)-(40,192),1,BF
300 LINE (210,0)-(256,192),1,BF
310 X=120:Y=0:FI=1
320 TIME=0
330 YB=180
340 GOSUB 690
350 ON SPRITE GOSUB 640:SPRITE ON
360 Y1=Y1+2:YB=YB-4
370 Y=Y1MOD 192
380 FF= YB+R
390 XP=XB-30
400 XA=XB+60


```



```


410 T=TIME/50
420 IF FF>209 THEN FF=209
430 PUTSPRITE 1,(X,Y),1,FI
440 IF XP<50 THEN XP=XB+60
450 PUTSPRITE 2,(XP,FF-20),12,2
460 IF XA>185 THEN XA=XB-30
470 IF XA=XP THEN 490
480 PUTSPRITE 8,(XA,YB+20),12,2
490 PUTSPRITE 3,(XB,YB),4,3
500 PUTSPRITE 6,(XC,FF),4,3
510 GOSUB 560
520 PUTSPRITE 4,(XB+35,YB+20),6,3
530 PUTSPRITE 7,(XC+35,FF+20),6,3
540 IF T>100 THEN 740
550 GOTO 360
560 D=STICK(CT)
570 IF D=3 THEN FI=5:X=X+4
580 IF D=7 THEN FI=1:X=X-4
590 IF YB+R-10<0 THEN GOSUB 690
600 SPRITE ON
610 IF X>XC AND X+16<XC+35 AND Y+16>FF AND Y<FF+36 AND FI<>4 THEN PU=PU+50
620 IF X>XB AND X+16<XB+35 AND Y+16>YB AND Y<YB+36 AND FI<>4 THEN PU=PU+50
630 RETURN
640 FI=4
650 SPRITE OFF
660 TI=TI+10
670 BEEP
680 RETURN
690 XB=INT(RND(1)*100)+50
700 XC=INT(RND(1)*100)+50
710 R=INT(RND(1)*60)+50
720 YB=193
730 RETURN
740 COLOR 1,15,4:SCREEN 1
750 LOCATE 2,5:PRINT"HAS TERMINADO TU PRUEBA"
760 TI=T+TI
770 LOCATE 3,8:PRINT"TIEMPO="TI;"SEG"
780 LOCATE 3,11:PRINT"PUNTOS="PU
790 LOCATE 2,18:PRINT"OTRA PARTIDA SI O NO [S/N]"
800 G$=INKEY$
810 IF G$="S" OR G$="s" THEN CLEAR:GOTO 270
820 IF G$="N" OR G$="n" THEN END
830 GOTO 800
840 DATA 1,2,2,1,7,F,1B,31,33,22,22,22,23,22,4,9,80,40,40,80,E0,F0,D8,8C,CC,48,4
4,D2,61,40,80,0,0,1,0,7,F,0,F,1F,0,1F,3F,0,0,0,0,0,80,C0,80,F0,F8,80,F8,FC,80,FC
,FE,C0,C0,C0,C0,C0,1F,1F,1F,1F,1F,10,10,10,10,10,10,10,10,0,0,F8,F8
850 DATA F8,F8,F8,8,8,8,8,8,8,8,8,0,0,0,7F,2,2,A,12,22,73,B1,1B,F,7,1,2,2,1,40,
E0,10,8,1C,32,61,C0,80,DE,DF,E1,81,41,41,81
860 DATA 1,2,2,1,7,F,1B,31,33,12,22,4B,86,2,1,0
870 DATA 80,40,40,80,E0,F0,D8,8C,CC,44,44,44,C4,44,20,90,1,2,2,1,7,F,1B,31,33,12
,22,4B,86,2,1,0,

```

DEFENDER

```
10 '00000000000000000000
20 '0000 0000
30 '0000 DEFENDER 0000
40 '0000 0000
50 '00000000000000000000
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 ' 1986
100 '00000000000000000000
110 '
120 COLOR 15,1,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(5,80),6:PRINT#1,"DEFENDER"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1:KEY OFF
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR{J/C}"
210 X$=INKEY$
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X$="J"OR X$="j" THEN CT=1:GOTO270
250 IF X$="c"OR X$="C" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 15,1,10:SCREEN 0
280 OPEN"GRP:"AS#1
290 SCREEN 2, 2 :FOR K=1 TO 3 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$
)):NEXT:SPRITE$(K)=B$:NEXT
300 Y1=10:Y2=40:Y3=70:Y4=100:Y5=130:Y6=160
310 A=2:B=6:GS=6:Y=90
320 TIME=0
330 GOSUB 700
340 ON SPRITE GOSUB 470
```

RALLY

```

10 '000000000000000000000000
20 '0000      0000
30 '0000      RALLY      0000
40 '0000      0000
50 '000000000000000000000000
60 '
70 'EUGENIO GARRIDO GOMEZ
80 '
90 '      1986
100 '000000000000000000000000
110 '
120 RESTORE:COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(30,80),6:PRINT#1," RALLY "
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR{J/C}"
210 X$=INKEY$
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X$="J"OR X$="j" THEN CT=1:GOTO270
250 IF X$="c"OR X$="C" THEN CT=0:GOTO270
260 GOTO 210
270 NI=9
280 SCREEN 1,2:COLOR 15,4,1:A$="||||||||||||||||||||||||||||||||"
290 VPOKE 8192+26,10:VPOKE8192+27,(6*16)+4
300 GOSUB 520:GOSUB 500:GOSUB 640:PO=4:KL=80:CO=1
310 KEY OFF
320 IF CO=1 THEN A=INT(RND(1)*20)ELSE A=40
330 IF A>=0AND A<10 THEN PO=PO+1
340 IF A>10 AND A<20 THEN PO=PO-1


```




```

350 IF PO<0 THEN PO=0
360 IF PO+NI>32 THEN PO=32-NI
370 IF NI<3 THEN CO=0
380 DIS=STICK(CT)
390 IF DIS=7 THEN KL=KL-8
400 IF DIS=3 THEN KL=KL+8
410 IF A=40AND SCR=20 THEN GOTO 560
420 IF SCR=40 THEN PLAY"ac":NI=NI-1:GOSUB 500:SCR=0
430 LOCATE 0,22:PRINTA$
440 LOCATE PO,22:PRINTB$
450 LOCATE 0,23:PRINT
460 PUT SPRITE 0,(KL,20),1,1
470 HJ=INT(KL/8)+6240
480 IF VPEEK(HJ)<>32 THEN GOTO 650
490 SCR=SCR+1:GOTO 320
500 B$=" "+SPACE$(NI)+" "
510 RETURN
520 FOR K=1 TO 3 :C$="":FORI=1TO 32 :READ H$:C$=C$+CHR$(VAL("&H"+H$)):NEXT:SPRIT
E$ (K)=C$:NEXT
530 DATA3A,3B,3F,3A,37,6,6,7,5,6,7,2,1B,1F,1B,1,5C,DC,FC,5C,EC,60,60,E0,A0,60,E0
,40,D8,F8,DB,80,0,0,0,1,8,24,25,93,4E,2A,9D,62,AC,6B,BE,DB,0,0,0,0,90,A4,24,C9,F
2,B4,29,DE,35,AA,DF,A9,0,0,0,0,0,0,3D,45,45,CD,FD,FD,FF,CF,4B,30,0,0
540 DATA0,0,0,FF,FF,F7,E3,F7,FF,FF,FF,FF,F2,D2,C
550 RETURN
560 FOR N=1 TO 20
570 LOCATE 0,22:PRINTA$
580 LOCATE PO,22:PRINTB$
590 LOCATE 0,23:PRINT
600 NEXT N
610 LOCATE PO,22:PRINT" META "
620 FOR N=1 TO 4:LOCATE 0,23:PRINT:NEXT N
630 FOR N=1TO 200:NEXTN:GOTO 10
640 CLS:FOR N=1 TO 22:PRINT"++++"
650 GOSUB 680:PUT SPRITE 0,(KL,20):FOR N=1 TO20:PUT SPRITE 1,(KL,20),9,2:VPOKE
8192+26,6:FOR M=1TO20:NEXTM:VPOKE 8192+26,10:PUT SPRITE1,(KL,20),1,2:NEXT N:GOSU
B 690
660 PLAY"ADADADADADADADAD":FOR N=255 TO 0 STEP-1:PUT SPRITE 2,(N,20),6,3:FOR M=1
TO 10:NEXT M:NEXT N
670 SCREEN1:LOCATE 0,10:PRINT"LO SIENTO TE HAS ESTRELLADO":FOR N=1 TO 5000:NEXTN
:GOTO 10
680 SOUND 6,&HF:SOUND7,&H7:SOUND8,&H10:SOUND9,&H10:SOUND10,&H10:SOUND12,&H40:SOU
ND 13,&H0:RETURN
690 FOR N=0 TO 13:SOUNDN,0:NEXTN:RETURN

```





SIETE Y MEDIA


```
10 KEY OFF:WIDTH 39:COLOR 15,4,4:SCREEN 0
20 INPUT "Como te llamas":N$
30 IF LEN (N$)>10 GOTO 20
40 OPEN "GRP:"AS#1
50 SCREEN2:COLOR 15,12,12:CLS
60 C=INT(RND(-TIME)*2):DIM A$(9),B$(4),C$(9,3)
70 DATA A,2,3,4,5,6,7,J,Q,K,E,F,C,D
80 FOR I=0 TO 9:READ A$(I):NEXT I:FOR I=0 TO 3:READ B$(I):NEXT I
90 PA=1:H=0:S=0:IH=8:IS=8:TS=0:TH=0:ERASE C$:DIM C$(9,3)
100 LINE (0,0)-(255,9),1,BF:DRAW"BM6,1":PRINT #1," # 7 y media # J. Suárez 1985"
110 LINE (0,24)-(72,33),1,BF:DRAW"BM1,25":PRINT#1,"ORDENADOR":LINE (120,24)-(180,33),1,BF:DRAW"BM120,25":PRINT#1,PS
120 LINE (0,96)-(LEN(N$)*8,105),1,BF:DRAW"BM1,97":PRINT#1,N$:LINE(120,96)-(180,105),1,BF:DRAW"BM120,97":PRINT#1,PH
130 LINE(0,40)-(255,88),12,BF:LINE(0,112)-(255,160),12,BF
140 X=8:Y=40:GOSUB 9000:Y=112:GOSUB 9000
150 IF C=1 GOTO 1000
160 IF H=1 GOTO 5000
170 H=1
180 GOSUB 9030
190 GOSUB 9050
195 CH=CO:XH=XX:YH=YY
200 Y=112:X=IH:GOSUB 9000:GOSUB 9100
210 IF XX<8 THEN OO=XX:GOTO 230
220 OO=.5
230 IF TH+OO>7.5 THEN GOSUB 9090:PA=0:GOTO 5000
240 GOSUB 9040
250 Z$=INKEY$:IF Z$="" OR (Z$<>"1" AND Z$<>"2") GOTO250
260 IF Z$="2" THEN PA=1:GOTO 5000
270 TH=TH+OO:GOSUB 9200:IH=IH+24:GOTO 190
1000 IF S=1 GOTO 5000
1010 S=1:GOSUB 9020
1020 GOSUB 9050
1025 CS=CO:XS=XX:YS=YY
1030 Y=40:X=IS:GOSUB 9000
1040 IF XX<8 THEN TS=TS+XX:GOTO 1060
1050 TS=TS+.5
1060 IF TS>7.5 THEN GOSUB 9140:GOSUB 9200:GOTO 5000
1065 IF TS=.5 GOTO 4000
1070 IF PA=0 GOTO 1110
1080 IF C=1 GOTO 1100
1090 IF TH>TS OR (TH=TS AND C=0) GOTO 4000
1095 IF TH=TS AND C=1 GOTO 1110
1096 GOTO 1120
1100 IF RND(-TIME)>.2 GOTO 1120
1110 GOSUB 9300:GOTO 5000
1120 IF TS>6.5 GOTO 1110
```



```

1130 IF TS>5.5 AND RND(-TIME)>.1 GOTO 1110
1140 IF TS>5.5 GOTO 4000
1150 IF TS>5 AND RND(-TIME)>.2 GOTO 1110
1160 IF TS>5 GOTO 4000
1170 IF TS>4 AND RND(-TIME)>.4 GOTO 1110
1180 IF TS>4 GOTO 4000
1190 IF TS>2 AND RND(-TIME)>.6 GOTO 1110
1200 IF TS>2 GOTO 4000
1210 IF TS<=2 AND RND(-TIME)>.8 GOTO 1110
4000 GOSUB 9200:IS=IS+24:GOTO 1020
5000 IF H=0 GOTO 160
5010 IF S=0 GOTO 1000
5020 Y=40:X=IS:CO=CS:XX=XS:YY=YS:GOSUB 9200
5030 Y=112:X=IH:CO=CH:XX=XH:YY=YH:GOSUB 9200
5040 IF PA=0 AND TS>7.5 GOTO 90
5050 IF PA=0 GOTO 6000
5055 IF TS>7.5 GOTO 7000
5060 IF TS>TH+00 GOTO 6000
5070 IF TS=TH+00 AND C=1 GOTO 6000
5080 GOTO 7000
6000 GOSUB 9310:PS=PS+TS:GOTO 8000
7000 GOSUB 9320:PH=PH+TH+00
8000 GOSUB 9010:BEEP:DRAW"BM 4,176 C1":PRINT#1,"[1] OTRA MANO [2] OTRO PROGRAMA"
8010 Z$=INKEY$:IF Z$="" OR (Z$<>"1" AND Z$<>"2") GOTO 8010
8020 IF Z$="2" THEN COLOR 15,4,4:SCREEN0:LOCATE4,10:PRINT"PARA EL SIGUIENTE PROG
RAMA":LOCATE 4,12:PRINT"PULSA PLAY EN EL CASSETTE":MOTOR ON:CLOAD
8100 IF C=1 THEN C=0:GOTO 90
8200 C=1:GOTO 90
9000 LINE(X,Y)-(X+24,Y+40),8,BF:FOR I=0 TO 4:N=I*8+Y:DRAW"C 15 BM=X; ,=N; ":PRINT#
1,"|||":NEXT I:RETURN
9010 LINE(0,168)-(255,191),1,BF:RETURN
9020 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"MI TURNO":FOR I=1 TO 400:NEXT
I:RETURN
9030 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"TU TURNO":FOR I=1 TO 400:NEXT
I:RETURN
9040 GOSUB 9010:BEEP:DRAW"C1 BM 10,176":PRINT#1,"[1] OTRA CARTA [2] TE PLANTAS"
:FOR I=1 TO 400:NEXT I:RETURN
9050 XX=INT(1+RND(-TIME)*10):YY=INT(1+RND(-TIME)*4):CO=8:IF YY<3 THEN CO=1
9060 IF C$(XX-1,YY-1)="1" GOTO 9050
9070 C$(XX-1,YY-1)="1":RETURN
9090 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"LO SIENTO":FOR I=1 TO 400:NEXT
I:RETURN
9100 LINE(X,Y)-(X+16,Y+16),15,BF:DRAW"BM =X; ,=Y; ":COLOR CO:PRINT#1,A$(XX-1):K=Y+
8:DRAW"BM =X; ,=K; ":PRINT#1,B$(YY-1):COLOR 15:RETURN
9140 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"ME HE PASADO":FOR I=1 TO 400:N
EXT I:RETURN
9200 LINE(X,Y)-(X+24,Y+40),15,BF:DRAW"BM =X; ,=Y; ":COLOR CO:PRINT#1,A$(XX-1):K=Y+
8:DRAW"BM =X; ,=K; ":PRINT#1,B$(YY-1):COLOR 15:RETURN
9300 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"ME PLANTO":FOR I=1 TO 400:NEXT
I:RETURN
9310 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"HE GANADO":FOR I=1 TO 400:NEXT
I:RETURN
9320 GOSUB 9010:BEEP:DRAW"C1 BM 100,176":PRINT#1,"HAS GANADO":FOR I=1 TO 400:NEX
T I:RETURN

```

HISTOGRAMA


```
10 ' PROGRAMA <HISTOGRAMA>
20 '
30 COLOR 15,4,4
40 WIDTH 39:KEY OFF:SCREEN 0
50 LOCATE 14,2:PRINT"HISTOGRAMA"
60 LOCATE 2,6:PRINT"Este programa de utilidad permite la introducción de da
tos numricos para surepresentación gráfica en un histogramade frecuencias a
bsolutas."
70 PRINT:PRINT" El número máximo de datos que acepta el programa es superi
or a 1000 y puedenser introducidos en cualquier orden."
80 LOCATE 5,20:PRINT"PULSA UNA TECLÁ PARA CONTINUAR"
90 IF INKEY$="" THEN 90
100 DEF FN R(X)=INT(X*1000+.5)/1000
110 CLS
120 INPUT"¿N. DE DATOS";N
130 DIM X(N)
140 PRINT
150 FOR Z=1 TO N
160 PRINT"INTRODUCE EL DATO N.";Z;
170 INPUT X(Z)
180 IF Z=1 THEN MX=X(1):MN=MX
190 A=A+X(Z)
200 CX=CX+X(Z)*X(Z)
210 IF X(Z)>MX THEN MX=X(Z) ELSE IF X(Z)<MN THEN MN=X(Z)
220 NEXT Z
230 M=A/N
240 V=(CX-A*A/N)/(N-1)
250 DT=SQR(V)
260 PLAY"T25504L64ABABAB"
270 CLS
280 LOCATE 0,1:PRINT"RESULTADOS ESTADISTICOS:"
290 PRINT:PRINT"N. DE DATOS:";N
300 PRINT:PRINT"SUMA DE LOS DATOS:";A
310 PRINT:PRINT"MEDIA:";FN R(M)
320 PRINT:PRINT"VARIANZA:";FN R(V)
330 PRINT:PRINT"DESVIACION TIPICA:";FN R(DT)
340 PRINT:PRINT"VALOR MAXIMO:";MX
350 PRINT:PRINT"VALOR MINIMO:";MN
360 RN=MX-MN
370 PRINT:PRINT"RANGO:";RN
380 CE=INT((RN/12)*10+1)/10
390 PRINT:PRINT"TE RECOMIENDO UNA AMPLITUD PARA LOS INTERVALOS DE";CE
400 PRINT:INPUT"¿QUE AMPLITUD QUIERES";B
410 C=INT(RN/B+1)
420 IF C>12 THEN PRINT:PRINT"N. DE CLASES DEMASIADO ALTO, PRUEBA CONOTRA AM
PLITUD":GOTO 400
430 DIM N(C),A(C),M(C)
440 FOR Z=1 TO N
```



```

450 FOR X=1 TO C
460 IF X(Z)>=MN+(X-1)*B AND X(Z)<MN+X*B THEN N(X)=N(X)+1:A(X)=A(X)+X(Z)
470 IF N(X)>NM THEN NM=N(X)
480 NEXT X,Z
490 CLS
500 LOCATE 5,2:PRINT"DIVISION EN INTERVALOS":PRINT:PRINT
510 PRINT"CLASE";TAB(10);"INTERVALO";TAB(23);"DATOS";TAB(30);"MEDIA"
520 FOR X=1 TO C
530 IF N(X)=0 THEN M(X)=0 ELSE M(X)=A(X)/N(X)
540 PRINT TAB(1);USING"###";X;
550 PRINT TAB(7);USING"####.##";MN+(X-1)*B;:PRINTTAB(13);"-";TAB(14);USING"###.##";MN+X*B;
560 PRINTTAB(23);USING"####";N(X);
570 IF N(X)=0 THEN PRINTTAB(30);"-----" ELSE PRINTTAB(29);USING"####.##";M(X)
580 NEXT X
590 LOCATE 2,21:PRINT"PARA EL HISTOGRAMA PULSA UNA TECLA"
600 IF INKEY$="" THEN 600 ELSE CLS
610 LOCATE 4,10:PRINT"CUANDO QUIERAS DEJAR DE VER EL":LOCATE 8,12:PRINT"HISTOGRAMA PULSA TECLA"
620 FOR T=1 TO 2000:NEXT
630 SCREEN 2
640 IF NM<10 THEN RS=1 ELSE IF NM<100 THEN RS=10 ELSE RS=100
650 Q=INT(INT(NM/RS+1)/2*RS+.5)/10
660 OPEN"GRP:" AS #1
670 DRAW"BM12,0":PRINT#1,"N"
680 DRAW"BM30,5C15S4A0D171R225"
690 DRAW"BM240,184":PRINT#1,"C"
700 FOR X=8 TO 168 STEP 16
710 NI=(168-X)/8*Q:XX=X+4
720 IF NI<>INT(NI) THEN 750
730 DRAW"BM4,=XX;":PRINT#1,USING"###";NI
740 LINE(28,X+8)-(31,X+8)
750 NEXT
760 FOR X=1 TO C
770 XX=14+X*16:X1=XX+4
780 DRAW"BM=X1;,182":PRINT#1,X
790 LINE(XX+16,175)-(XX+16,178)
800 NEXT X
810 FOR Z=1 TO C
820 IF N(Z)=0 THEN 870
830 FOR X=0 TO INT(N(Z)/Q-.5)
840 X1=23+Z*16:X2=X1+6:YY=168-X*8
850 DRAW"BM=X1;,=YY;":PRINT#1,"█":DRAW"BM=X2;,=YY;":PRINT#1,"█"
860 NEXT X
870 NEXT Z
880 IF INKEY$="" THEN 880 ELSE CLOSE#1
890 SCREEN 0
900 LOCATE 2,4:PRINT"ELIGE UNA OPCION:"
910 LOCATE 4,7:PRINT"1.- INTRODUCIR OTROS DATOS"
920 LOCATE 4,9:PRINT"2.- DEFINIR OTROS INTERVALOS"
930 LOCATE 4,11:PRINT"3.- FIN DEL PROGRAMA"
940 O$=INPUT$(1)
950 IF O$<>"1" AND O$<>"2" AND O$<>"3" THEN 940
960 IF O$="1" THEN RUN 100 ELSE IF O$="2" THEN ERASEN,M,A:GOTO 270
970 CLS
980 LOCATE 4,10:PRINT"PARA EL SIGUIENTE PROGRAMA"
990 LOCATE 4,12:PRINT"PULSA PLAY EN EL CASSETTE"
1000 CLOAD
1010 GOTO 1000 .

```

DIANA


```
10 * PROGRAMA <DIANA>
20 *
30 COLOR4,15,4
40 KEY OFF
50 SCREEN0
60 LOCATE 14,0:PRINT"DIANA"
70 PRINT:PRINT" En este programa te transformas en el ":PRINT"valiente pil
oto de un carro de combate ":PRINT"en prácticas de tiro sobre una diana
":PRINT"móvil."
80 PRINT:PRINT" Por cada disparo acertado ganarás 10 ":PRINT"puntos, pero
cada vez que falles perde- ":PRINT"rás 1 punto. Tienes hasta un total de
":PRINT"10 pantallas en las que aumenta el nú- ":PRINT"mero de obstáculo
s."
90 LOCATE 5,20:PRINT"PULSA UNA TECLA PARA LAS":LOCATE5,22:PRINT"INSTRUCCION
ES DE MANEJO."
100 IF INKEY$="" THEN 100 ELSE CLS
110 LOCATE2,5:PRINT" El tanque se mueve arriba y abajo ":PRINT"con las te
clas de cursor.":PRINT:PRINT" El disparo se consigue pulsando la":PRINT:PR
INT"barra espaciadora."
115 LOCATE 0,15:INPUT"JUEGAS CON TECLADO (0) O JOYSTICK (1) ";DS
120 LOCATE1,20:PRINT"PULSA UNA TECLA PARA COMENZAR EL JUEGO"
130 IF INKEY$="" THEN 130
140 COLOR10,15,4
150 OPEN"GRP:"AS#1
160 SCREEN 2
170 GOSUB 780
180 SC=1
190 LINE(0,0)-(20,191),4,BF:LINE(235,0)-(255,191),4,BF:LINE(0,0)-(255,14),4
,BF
200 GOSUB 720:GOSUB760
210 VL=0:DI=0
220 IF SC>10 THEN GOTO 420
230 FOR Z=1 TO 11+SC
240 PS=INT(RND(1)*22+2)*8
250 IF POINT(121,PS)=10 THEN 240
260 DRAW"BM118,=PS;C12R6D1L5D1L1R6BM130,=PS;R6L1D1L5D1R6"
270 PS=PS+4
280 DRAW"BM118,=PS;R6D1L5D1L1R6BM130,=PS;R6L1D1L5D1R6"
290 NEXT
300 XT=40
310 YT=INT(RND(1)*165+16)
320 GOSUB 700
330 YD=INT(RND(1)*165+16)
340 GOSUB660
350 ON STRIG GOSUB 510,510:STRIG(DS) ON
360 ON INTERVAL=20 GOSUB660:INTERVAL ON
370 D=STICK(DS)
```



```

380 IF D=1 THEN D=1 ELSE IF D=5 THEN D=2 ELSE 370
390 ON D GOSUB 600,610
400 GOTO 370
410 GOTO 410
420 FOR T=1 TO 1000:NEXT
430 SCREEN0
440 LOCATE4,8:PRINT"PULSA J PARA OTRO JUEGO":LOCATE4,10:PRINT"PULSA T PARA
TERMINAR"
450 R$=INPUT$(1)
460 IF R$="" THEN 450 ELSE IF R$="J" OR R$="j" THEN RUN 140 ELSE IF R$<>"T"
AND R$<>"t" THEN 450
470 CLS
480 LOCATE4,10:PRINT"PARA EL SIGUIENTE PROGRAMA"
490 LOCATE4,12:PRINT"PULSA PLAY EN EL CASSETTE"
500 CLOAD
510 STRIG(0) OFF:INTERVAL OFF
520 ON SPRITE GOSUB 620:SPRITE ON
530 PUT SPRITE 0,(XT,YT),6,0
540 FOR X=XT+3 TO 230 STEP 3
550 PUT SPRITE4,(X,YT),8,4
560 IF POINT(X+6,YT+4)=12 THEN X=255:VL=0:PLAY"t25502L26"
570 NEXT
580 PUT SPRITE4,(0,200),4,4
590 IF VL<>1 THEN STRIG(0)ON:INTERVAL ON:PT=PT-1:GOSUB 720:RETURN ELSE PLAY"
o4164cdedefdefcdd":SC=SC+1:PT=PT+10:LINE(21,15)-(234,191),15,BF:RETURN 190
600 YT=YT-1*ABS(YT>14):PUT SPRITE 0,(40,YT),6,2:RETURN
610 YT=YT+1*ABS(YT<183):PUT SPRITE 0,(40,YT),6,1:RETURN
620 X=255:VL=1
630 PUT SPRITE0,(-10,YT),4,0
640 PUT SPRITE3,(-10,YD),4,3
650 RETURN
660 S=RND(1):IF S>.5 THEN SG=5 ELSE SG=-5
670 YD=YD+SG:IFYD<22 OR YD>178 THEN SG=-SG:GOTO 670
680 PUT SPRITE 3,(215,YD),1,3
690 RETURN
700 PUT SPRITE 0,(XT,YT),6,0
710 RETURN
720 DRAW"BM30,0":PRINT#1,"Puntos:"
730 LINE(86,0)-(120,10),4,BF
740 DRAW"bm88,0":PRINT#1,USING"###";PT
750 RETURN
760 DRAW"bm150,0":PRINT#1,"Diana: ";SC
770 RETURN
780 RESTORE 870:FOR SP=0 TO 4
790 T$=""
800 FOR Z=1 TO 8
810 READ E
820 T$=T$+CHR$(E)
830 NEXT
840 SPRITE$(SP)=T$
850 NEXT
860 RETURN
870 DATA 252,252,120,79,79,120,252,252,195,255,231,231,255,219,24,24,24,24,
219,255,231,231,255,195,28,24,56,56,56,56,24,28,0,0,0,3,3,0,0,0

```

SOLITARIO

```

1 REM Copyright 1985 by J.L.T. XV.
2 WIDTH39:KEYOFF:COLOR 15,4,4:CLS:RESTORE100:LOCATE17,0:PRINT"SOLO":FORI=1
TO17:READX$:LOCATE0,I+1:PRINTX$:NEXTI
3 LOCATE0,20:PRINT"      PULSA TECLA PARA CONTINUAR":X$=INKEY$:IFX$=""GOTO 3
4 CLS:FORI=1TO9:READX$:LOCATE0,I+1:PRINTX$:NEXTI
5 LOCATE0,20:PRINT"      PULSA TECLA PARA COMENZAR":X$=INKEY$:IFX$=""GOTO 5
10 DIMT$(15),C(7,7):CLS
100 CLS:GOSUB3000
200 LOCATE0,18:PRINT"DESDE FILA: ";:BEEP
201 X$=INKEY$:IFX$=""THEN201ELSEIFX$="O"THEN5000ELSEF1=VAL(X$)
202 LOCATE11,18:PRINTF1:LOCATE20,18:BEEP:PRINT"COLUMNA: "
203 X$=INKEY$:IFX$=""THEN203ELSEIFX$="O"THEN5000ELSEC1=VAL(X$)
206 LOCATE28,18:PRINTC1:LOCATE0,19:PRINT"HASTA FILA: ";:BEEP
208 X$=INKEY$:IFX$=""THEN208ELSEIFX$="O"THEN5000ELSEF2=VAL(X$)
210 LOCATE11,19:PRINTF2:LOCATE20,19:BEEP:PRINT"COLUMNA: "
212 X$=INKEY$:IFX$=""THEN212ELSEIFX$="O"THEN5000ELSEC2=VAL(X$):LOCATE11,19
213 LOCATE28,19:PRINTC2
214 IF(F1<1)OR(F1>7)OR(C1<1)OR(C1>7)OR(F2<1)OR(F2>7)OR(C2<1)OR(C2>7)GOTO 23
0
216 IF(C(F1,C1)<>1)OR(C(F2,C2)<>-1)OR(C((F1+F2)/2,(C1+C2)/2)<>1)GOTO 230
218 IF((F1=F2)AND(ABS(C1-C2)=2))OR((C1=C2)AND(ABS(F1-F2)=2))THEN220ELSE230
220 C(F1,C1)=-1:LOCATEC1*2,F1*2:PRINT"  ":C((F1+F2)/2,(C1+C2)/2)=-1:LOCATE(C
1+C2),(F1+F2):PRINT"  ":C(F2,C2)=1:LOCATEC2*2,F2*2:PRINT"  ":BEEP
230 GOSUB3100:GOTO 200
1000 DATA"  1 2 3 4 5 6 7 "
1001 DATA"      XWRWRWY      ","1      V V V V      ","      TWUWUWS      ","2      V V
V V      ","      XWRWUWUWUWRWY"
1002 DATA"3V V V V V V V V"," TWUWUWUWUWUWUWS","4V V V V V V V V"," TWUWUWU
WUWUWUWS","5V V V V V V V V"
1003 DATA " ZWQWUWUWUWUWQW","6      V V V V      ","      TWUWUWS      ","7      V
V V V      ","      ZWQWQW"
3000 RESTORE100:FORI=0TO15:READJ$:T$(I)=J$:NEXTI

```



```

3002 FORI=0TO15:LOCATE0,I:PRINT$(I):NEXTI
3010 FORI=3TO5:FORJ=1TO7:C(I,J)=1:LOCATEJ*2,I*2:PRINT"█":NEXTJ:NEXTI
3012 FORI=3TO5:FORJ=1TO7:C(J,I)=1:LOCATEI*2,J*2:PRINT"█":NEXTJ:NEXTI:C(4,4)
=-1:LOCATE8,8:PRINT"  ":RETURN
3100 FORK=18TO21:LOCATE0,K:PRINT"  ":NEXTK
XTK:RETURN
5000 GOSUB3100:LOCATE0,19:PRINT"  E:EMPEZAR S:SOLUCIONAR F:FINALIZAR"
5002 X$=INKEY$:IFX$=""THEN5002ELSEIF(X$="E")OR(X$="e")THEN100ELSEIF(X$="S")
OR(X$="s")THEN8000ELSEIFX$="f"ORX$="F"THEN5500ELSE5002
5500 COLOR 15,4,4:CLS:LOCATE7,10:PRINT"PARA EL PROXIMO PROGRAMA":LOCATE14,1
2:PRINT"PULSA PLAY":CLOAD
8000 CLS:RESTORE9000:GOSUB3000
8002 FORI=1TO31:READF1:READC1:LOCATE0,18:PRINT"DESDE  FILA: ";F1:BEEP:LOCAT
E20,18:PRINT"COLUMNA: ";C1:BEEP
8004 READF2:READC2:LOCATE0,19:PRINT"HASTA  FILA: ";F2:BEEP:LOCATE20,19:PRIN
T"COLUMNA: ";C2:BEEP:FORJ=0TO300:NEXTJ
8006 LOCATEC1*2,F1*2:PRINT"  ":LOCATE(C1+C2),(F1+F2):PRINT"  ":LOCATEC2*2,F2*
2:PRINT"█":BEEP:FORJ=0TO300:NEXTJ:GOSUB3100:NEXTI:GOTO 5000
9000 DATA6,4,4,4,5,2,5,4,3,2,5,2,5,1,5,3,3,1,5,1,5,4,5,2,5,1,5,3,5,6,5,4,7,
5,5,5,7,3,7,5,4,5,6,5,7,5,5,5,5,4,5,6,2,5,4,5,2,3,2,5,1,5,3,5,1,3,1,5,4,5,2
,5,1,5,3,5
9002 DATA4,7,4,5,5,7,5,5,4,5,2,5,3,7,3,5,2,5,4,5,5,5,3,5,4,3,2,3,6,3,4,3,4,
4,4,2,3,5,3,3,2,3,4,3,4,2,4,4
9100 DATA" SOLO  es un solitario, que normalmente ", "se juega con monedas,
sobre un tablero ", "de 7x7 sin las esquinas.", " "
9102 DATA" En cada uno de los cuadros hay una mo-", "neda, salvo en el centr
al, que está li-", "bre. El objetivo del solitario es in-", "vertir la situac
ión, es decir ,dejar el"
9104 DATA"tablero vacío, salvo el cuadro central.", " ", " Se pueden mover l
as fichas, haciendo", "saltar una por encima de otra contigua", "(a su dere
cha o a su izquierda, arriba"
9106 DATA"o abajo) si es que al otro lado de sta", "hay un cuadro vacío. En
tonces se come", "aquella sobre la que se ha saltado y se", "la retira del t
ablero."
9110 DATA" El programa primero te pedirá la fila ", "y la columna de la fich
a que vas a mo-", "ver y a continuación la fila y la co-", "lumna del cuadro
donde la quieres colo-", "car.
9112 DATA", " Para parar pulsa cero (0) cuando te", "pida fila o columna."

```



MSX

Nº 16

690 Ptas.

SOFTWARE

HUNGRY HARRY



CURSO DE **BASIC** + MICROORDENADORES

prácticas con...

**Microordenador
ZX SPECTRUM**



**Microordenador
COMMODORE**



**Microordenadores
AMSTRAD, MSX, PC**



Para saber cómo hablar con los ordenadores

El Curso CEAC a Distancia,
BASIC + Microordenadores,
le va a introducir paso
a paso, con un cuidado
método, en uno de los temas más
apasionantes de nuestros días:

la programación de ordenadores.

Al aprender PRACTICANDO desde un principio
a programar BASIC, lenguaje diseñado
especialmente para dar los primeros pasos
en programación, estará sentando las bases
para el estudio de cualquier otro
lenguaje de alto nivel.

**Curso CEAC
de BASIC + Microordenadores:
un diálogo permanente
con el ordenador.**

CEAC

CENTRO DE ENSEÑANZA A DISTANCIA
AUTORIZADO POR EL MINISTERIO DE
EDUCACIÓN Y CIENCIA N.º 8039185

(BOLETIN OFICIAL DEL ESTADO 3-6-83)
Aragón, 472 (Dpto. 08013 Barcelona
Tel.: (93) 245 33 06

ESTAS ENSEÑANZAS SE AJUSTAN AL ART. 35
DEL DECRETO 707/1976 Y A LA ORDEN MINISTERIAL DE 5/2/1979

Actúe ahora
en su propio
beneficio
y pídasenos
información.

GRATUITAMENTE

Sí, deseo recibir a la mayor
brevedad posible información
sobre el Curso de: _____

Nombre y apellidos _____ Edad _____

Domicilio _____

_____ Nº _____ Piso _____ Pta. _____ Tel. _____

C. Postal _____ Población _____

_____ Provincia _____

Profesión _____

CEAC. Aragón, 472
(Dpto.) 08013 Barcelona

o llame...
(93) 245 33 06
de Barcelona

